clear all
set more off
set mem 10000000
set matsize 10000
version 15

************************************************************
*** RDROBUST covariate smoothness, 2001 Census variables ***
************************************************************

** Set file paths
do "$path_code/paths.do"

** Set graph scheme
cd "$path/code/analyze"
set scheme fb, perm

****************************************************************** 
****************************************************************** 

{
use "$panel/panel_dataset_full.dta", clear

	// Keep villages in RD sample
gen in_fs_sample = vplan4<11 & corr_state==1 & sample==1 & sample_h==1 & sing_h==1 & pop_non_zero==1 
keep if in_fs_sample==1
	
	// Create state FEs (since RD robust doesn't let you pass them through)
drop if st_code==32 // Kerala, only 3 villages
tab st_code if in_fs_sample==1, gen(STFEfs)	
drop STFEfs1 // to avoid collinearity

	// Create district FEs (since RD robust doesn't let you pass them through)
gen stdtFE = stdt
tab stdtFE state     if inlist(stdtFE,62,63,64,70,91,175,180,181,202,308,309,320,490,493,494,495,504,505,508)	
replace stdtFE = 999 if inlist(stdtFE,62,63,64,70,91,175,180,181,202,308,309,320,490,493,494,495,504,505,508)
	// one catch-all district FE for districts with so few in-sample villages that they break rdrobust
tab stdtFE if in_fs_sample==1, gen(DTFEfs)	
drop DTFEfs1 // to avoid collinearity

	// Create block groups (for clustering)
egen stdtbk = group(stdt bk_code)

	// Create lights-difference variable, to identify crazy outliers
gen lights_diff = abs(lights_max2011_hat - lights_max2001_hat)

	// Create macros for lists of outcome variables 
global vars_demo = "pct_06_01 lit_p_01 pct_scst_01 "
global vars_labor_pct = "work_pooled_ag_m_01 work_pooled_ag_f_01 work_pooled_ot_m_01 work_pooled_ot_f_01 work_pooled_ag_p_01 work_pooled_ot_p_01 work_p_01 work_main_p_01 work_marg_p_01 work_m_01 work_main_m_01 work_marg_m_01 work_f_01 work_main_f_01 work_marg_f_01 "
global vars_vd = "vd_pwr_d_any_01 vd_pwr_d_dom_01 vd_pwr_d_agr_01 vd_pwr_d_all_01 vd_wat_d_any_01 vd_wat_d_tubewell_01 pct_irr_01 vd_geo_k_town_01 "

	// Create index of 7 VD 2001 education variables
foreach v of varlist vd_edu_n_*_01 {
	sum `v' if tot_p<=1000
	gen Z`v' = (`v' - r(mean))/(r(sd))
}
egen double index_vd_edu01 = rmean(Z*)
drop Z*	

	// Create index of 8 VD 2001 health variables
foreach v of varlist vd_hea_n_*_01 {
	sum `v' if tot_p<=1000
	gen Z`v' = (`v' - r(mean))/(r(sd))
}
egen double index_vd_hea01 = rmean(Z*)
drop Z*	

	// Create index of 4 VD 2001 financial variables
foreach v of varlist vd_fin_d_*_01 {
	sum `v' if tot_p<=1000
	gen Z`v' = (`v' - r(mean))/(r(sd))
}
egen double index_vd_fin01 = rmean(Z*)
drop Z*	

	// Create index of 4 VD 2001 communication variables
foreach v of varlist vd_com_d_*_01 {
	sum `v' if tot_p<=1000
	gen Z`v' = (`v' - r(mean))/(r(sd))
}
egen double index_vd_com01 = rmean(Z*)
drop Z*	

	// Create index of 4 VD 2001 transportation variables
foreach v of varlist vd_tra_d_*_01 vdpr_tra_d_app_pr_01 {
	sum `v' if tot_p<=1000
	gen Z`v' = (`v' - r(mean))/(r(sd))
}
egen double index_vd_tra01 = rmean(Z*)
drop Z*	

global vars_index = "index_vd_edu01 index_vd_hea01 index_vd_fin01 index_vd_com01 index_vd_tra01 "



	// Create variables to store regresson results
gen rf_step = .
gen yvar = ""
gen ifs = ""
gen controls_base = ""
gen control = ""
gen fe = ""
gen kernel = ""
gen bwmethod = ""
gen vce = ""
gen polynomial_order = .
gen beta_conv = .
gen beta_robust = .
gen se_conv = .
gen se_robust = .
gen pval_conv = .
gen pval_robust = .
gen lci_conv = .
gen uci_conv = .
gen lci_robust = .
gen uci_robust = .
gen bw_lo = .
gen bw_hi = .
gen nobs_orig = .
gen nobs_left = .
gen nobs_right = .
gen nobs_total = .
gen ndist = . 
gen ymean = .
gen ftag = ""
global h_wide = 200


	// Massive loop over RDROBUST 2001 covariates and sensitivities
foreach y in $vars_demo $vars_labor_pct $vars_vd $vars_index { 

	// Prep to store results
	foreach v of varlist rf_step-ftag {
		cap replace `v' = ""
		cap replace `v' = .
	}
	local row = 0

	// Define outcome-specific stuff
	global yvar = "`y'"
	global control = ""
	do "$path/code/analyze/RDROBUST_rf_outcomes_graphspecs.do"

	//Loop through sensitivities
	foreach rf_step in 1 2 3 4 5 7 8 9 10 11 {

		// Reset RDROBUST defaults
		local ifs = "in_fs_sample==1 & pop_mismatch20==0 & lights_diff<20"
		local controls_base = "" 
		local fe = "STFEfs*"
		local kernel = "tri"
		local bwmethod = "mserd"
		local vce = ""
		local poly = 1
		local graphtag = "fs"

		// Define step-specific RDROBUST settings
		if inlist(`rf_step',1) {
			local folder = "RDROBUST plots rf outcomes preferred"
			local ftag = "2001 covariates, preferred"
			local title2 = ""
		}
		if inlist(`rf_step',2) {
			local folder = "RDROBUST plots rf outcomes popmismatch"
			local ftag = "2001 covariates, popmismatch"
			local title2 = "(population mismatches)"
			local ifs = "in_fs_sample==1 & lights_diff<20"
		}
		if inlist(`rf_step',3) {
			local folder = "RDROBUST plots rf outcomes lights outliers"
			local ftag = "2001 covariates, lights outliers"
			local title2 = "(lights outliers)"
			local ifs = "in_fs_sample==1 & pop_mismatch20==0"
		}
		if inlist(`rf_step',4) {
			local folder = "RDROBUST plots rf outcomes epa kernel"
			local ftag = "2001 covariates, epa kernel"
			local title2 = "(Epanechnikov kernel)"
			local kernel = "epa"
		}
		if inlist(`rf_step',5) {
			local folder = "RDROBUST plots rf outcomes uni kernel"
			local ftag = "2001 covariates, uni kernel"
			local title2 = "(uniform kernel)"
			local kernel = "uni"
		}
		if inlist(`rf_step',6) {
			local folder = "RDROBUST plots rf outcomes no lights controls"
			local ftag = "2001 covariates, no lights controls"
			local title2 = "(no lights controls)"
			local controls_base = ""
		}
		if inlist(`rf_step',7) {
			local folder = "RDROBUST plots rf outcomes no FEs"
			local ftag = "2001 covariates, no FEs"
			local title2 = "(no fixed effects)"
			local fe = ""
		}
		if inlist(`rf_step',8) {
			local folder = "RDROBUST plots rf outcomes district FEs"
			local ftag = "2001 covariates, district FEs"
			local title2 = "(district fixed effects)"
			local fe = "DTFEfs*"
		}
		if inlist(`rf_step',9) {
			local folder = "RDROBUST plots rf outcomes nncluster by district"
			local ftag = "2001 covariates, nncluster by district"
			local title2 = "(nncluster by district)"
			local vce = "vce(nncluster stdt)"
		}
		if inlist(`rf_step',10) {
			local folder = "RDROBUST plots rf outcomes cluster by district"
			local ftag = "2001 covariates, cluster by district"
			local title2 = "(cluster by district)"
			local vce = "vce(cluster stdt)"
		}
		if inlist(`rf_step',11) {
			local folder = "RDROBUST plots rf outcomes CERRD bandwidth"
			local ftag = "2001 covariates, CERRD bandwidth"
			local title2 = "(CERRD bandwidth)"
			local bwmethod = "cerrd"
		}

		// Run first-stage regresssion
		di "rdrobust $yvar tot_p if `ifs', c(299.5) covs(`controls_base' $control `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'"
		rdrobust $yvar tot_p if `ifs', c(299.5) covs(`controls_base' $control `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'

		// Generate in-sample indicator and store bandwidths
		cap drop temp_in_reg
		qui gen temp_in_reg = `ifs' & inrange(tot_p,299.5-e(h_l),299.5+e(h_r)) & ${yvar}!=.
		local h_l = e(h_l)
		local h_r = e(h_r)

		// Store results
		local row = `row' + 1
		qui replace rf_step = `rf_step' in `row'
		qui replace yvar = "$yvar" in `row'
		qui replace ifs = "`ifs'" in `row'
		qui replace controls_base = "`controls_base'" in `row'
		qui replace control = "$control" in `row'
		qui replace fe = "`fe'" in `row'
		qui replace kernel = e(kernel) in `row'
		qui replace bwmethod = e(bwselect) in `row'
		qui replace vce = "`vce'" in `row'
		qui replace polynomial_order = e(p) in `row'
		qui replace beta_conv = e(tau_cl) in `row'
		qui replace beta_robust = e(tau_bc) in `row'
		qui replace se_conv = e(se_tau_cl) in `row'
		qui replace se_robust = e(se_tau_rb) in `row'
		qui replace pval_conv = e(pv_cl) in `row'
		qui replace pval_robust = e(pv_rb) in `row'
		qui replace lci_conv = e(ci_l_cl) in `row'
		qui replace uci_conv = e(ci_r_cl) in `row'
		qui replace lci_robust = e(ci_l_rb) in `row'
		qui replace uci_robust = e(ci_r_rb) in `row'
		qui replace bw_lo = e(h_l) in `row'
		qui replace bw_hi = e(h_r) in `row'
		qui replace nobs_orig = e(N) in `row'
		qui replace nobs_left = e(N_h_l) in `row'
		qui replace nobs_right = e(N_h_r) in `row'
		qui replace nobs_total = e(N_h_l) + e(N_h_r) in `row'
		qui unique stdt if temp_in_reg==1
		qui replace ndist = r(unique) in `row'
		qui sum $yvar if temp_in_reg==1 & tot_p<299.5
		qui replace ymean = r(mean) in `row'
		qui replace ftag = "`ftag'" in `row'

		// Residualize $yvar for regression sample
		qui reg $yvar `controls_base' $control `fe' if temp_in_reg==1
		cap drop y_resid_sample
		predict y_resid_sample, residuals
		
		// Residualize $yvar for [100,500] bandwidth
		qui reg $yvar `controls_base' $control `fe' if `ifs' & inrange(tot_p,299.5-${h_wide},299.5+${h_wide})
		cap drop y_resid_full
		predict y_resid_full, residuals

		// RD plots for regression sample (variable bandwidth) 
		foreach bins in 10 20 40 {
			if `h_l'<=75 {
				local xlab = "250 275 300 325 350"
			}
			else if inrange(`h_l',76,125) {
				local xlab = "200 250 300 350 400"
			}
			else if inrange(`h_l',126,175) {
				local xlab = "150 200 250 300 350 400 450"
			}
			else {
				local xlab = ""
			}
			rdplot y_resid_sample tot_p if temp_in_reg==1, ///
				c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_l' `h_r') ///
				graph_options( ///
				title("$title" "`title2'", color(black) size(large)) ///
				ytitle("$ytitle", size(medlarge)) ///
				xtitle("2001 village population", size(medlarge)) ///
				ylabel(,nogrid angle(0) labsize(medlarge)) ///
				xlabel(`xlab', labsize(medlarge)) ///
				graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
				legend(off))
			graph export "$results/`folder'/${yvar}_PRE_`graphtag'_inreg_`bins'.pdf", replace
		}

		if !inlist(`rf_step',11) {
			// RD plots for regression sample (constant bandwidth) 
			foreach bins in 20 40 {
				rdplot y_resid_full tot_p if `ifs' & inrange(tot_p,299.5-${h_wide},299.5+${h_wide}), ///
					c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(${h_wide} ${h_wide}) ///
					graph_options( ///
					title("$title" "`title2'", color(black) size(large)) ///
					ytitle("$ytitle", size(medlarge)) ///
					xtitle("2001 village population", size(medlarge)) ///
					ylabel(,nogrid angle(0) labsize(medlarge)) ///
					xlabel(, labsize(medlarge)) ///
					graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
					legend(off))
				graph export "$results/`folder'/${yvar}_PRE_`graphtag'_wide_`bins'.pdf", replace
			}
		}
	}

	// Save results
	preserve
	keep rf_step-ftag
	dropmiss, obs force
	if "$yvar"!="pct_06_01" {
		tempfile reg_results
		save `reg_results'
		clear
		append using "$results/RDROBUST_smoothness_census.dta" `reg_results'
	}
	duplicates drop
	compress
	save "$results/RDROBUST_smoothness_census.dta", replace
	restore
	
}



}

****************************************************************** 
****************************************************************** 

